


CRREL

REPORT 77-2



A computer program to determine the resistance of long wires and rods to nonhomogeneous ground



CRREL Report 77-2

A computer program to determine the resistance of long wires and rods to nonhomogeneous ground

Steven A. Arcone

January 1977

Prepared for

DIRECTORATE OF FACILITIES ENGINEERING
OFFICE, CHIEF OF ENGINEERS

By

CORPS OF ENGINEERS, U.S. ARMY

COLD REGIONS RESEARCH AND ENGINEERING LABORATORY
HANOVER, NEW HAMPSHIRE

Approved for public release; distribution unlimited.

REPORT DOCUMENTATION PAGE		READ INSTRUCTIONS BEFORE COMPLETING FORM
1. REPORT NUMBER CRREL Report 77-2	2. GOVT ACCESSION NO.	3. RECIPIENT'S CATALOG NUMBER
4. TITLE (and Subtitle) A COMPUTER PROGRAM TO DETERMINE THE RESISTANCE OF LONG WIRES AND RODS TO NONHOMOGENEOUS GROUND		5. TYPE OF REPORT & PERIOD COVERED
		6. PERFORMING ORG. REPORT NUMBER
7. AUTHOR(s) Steven A. Arcone		8. CONTRACT OR GRANT NUMBER(s)
9. PERFORMING ORGANIZATION NAME AND ADDRESS U.S. Army Cold Regions Research and Engineering Laboratory Hanover, New Hampshire 03755		10. PROGRAM ELEMENT, PROJECT, TASK AREA & WORK UNIT NUMBERS Project 4A762719AT33, Task 03, Work Unit 006
11. CONTROLLING OFFICE NAME AND ADDRESS Directorate of Facilities Engineering Office, Chief of Engineers Washington, D.C.		12. REPORT DATE January 1977
		13. NUMBER OF PAGES 20
14. MONITORING AGENCY NAME & ADDRESS (if different from Controlling Office)		15. SECURITY CLASS. (of this report) Unclassified
		15a. DECLASSIFICATION/DOWNGRADING SCHEDULE
16. DISTRIBUTION STATEMENT (of this Report) Approved for public release; distribution unlimited.		
17. DISTRIBUTION STATEMENT (of the abstract entered in Block 20, if different from Report)		
18. SUPPLEMENTARY NOTES		
19. KEY WORDS (Continue on reverse side if necessary and identify by block number) Electrical grounding Two-layer earth models Electrical resistivity Electrodes Resistance to ground		
20. ABSTRACT (Continue on reverse side if necessary and identify by block number) A computer program was developed for finding the d-c resistance to ground of two simple electrodes, a straight horizontal wire and a vertically driven rod. The objective of this study was to develop a rapid means of finding the resistance to ground of simple electrode types in arctic environments where a two-layer earth model, frozen and unfrozen ground, is applicable. The program can consider homogeneous as well as two-layer earth, and the length, diameter and position of the electrodes. The computations were performed first by dividing an electrode into several smaller segments. Next the electrostatic potential of each segment was computed at the center of the electrode for unit-applied current. The segment potentials were then summed to find the total resistance to ground. Some specific		

20. (cont'd)

computations are presented in comparison with previous theoretical work of other authors. The following conclusions were made: 1) A maximum run time of 165 seconds is needed for all two-layer arctic models where (a) the depth of the upper layer does not exceed 10 m, (b) the vertical rod length is less than 30 m, or (c) the horizontal wire length is less than 100 m; 2) Best accuracy is obtained when rod and wire radii are less than 0.01 m; and 3) Coincidence of the center of the vertical electrode with the two-layer interface must be avoided.

PREFACE

This report was prepared by Steven A. Arcone, Geophysicist, Physical Sciences Branch, Research Division, U.S. Army Cold Regions Research and Engineering Laboratory.

The study was performed under Project 4A762719AT33, *Research for Base Developments in Theaters of Operation*; Task 03, *Base Development in Winter Conditions*; Work Unit 006, *Site Selection and Subsurface Exploration*.

This report was technically reviewed by B. Pratt and P.V. Sellmann of CRREL.

The adaptation of mathematical solutions (eq 12-19) for the approach used was developed by Dr. Pieter Hoekstra, formerly Geophysicist of CRREL, and the computer programming was done by Kevin E. Gartner, Computer Technician, of CRREL.

SUMMARY

The purpose of this research was to develop an effective program for computing the resistance to ground of simple electrodes carrying direct current. A model of horizontal earth layers and parameters applicable to arctic engineering was employed. The report begins with a brief introduction to the theory of d-c earth conduction and presents the definition of *resistance to ground*. The development of the specific electrode cases of a horizontal straight wire and a vertically driven rod are then discussed. The explicit developments of the mathematical solutions are not presented because they may be found in the literature. However, the solutions themselves are presented and they are then numerically integrated using earth parameters based on previous studies of permafrost resistivity. The results of the simpler cases involving homogeneous ground compare favorably with those of previous theoretical studies.

The more difficult cases, involving two-layer earth models, generally require less than 165 sec of computer run time for layer thicknesses up to 10 m, rod lengths up to 30 m, and wire lengths up to 100 m. For these cases, best accuracy is obtained with electrode radii less than 0.01 m. Specific problems such as the coincidence of the two-layer interface with the electrode center and subsurface emplacement of a horizontal wire are also discussed.

The computer program developed, written in BASIC computer language is printed out in full in the appendix, and difficult cases involving lengthy run times and numerical inaccuracies are listed at the end of the report. No attempt was made to catalog results for variations in all the parameters that the program can consider. Instead, it is believed that the results presented are sufficient evidence for the capabilities of the program.

A COMPUTER PROGRAM TO DETERMINE THE RESISTANCE OF LONG WIRES AND RODS TO NONHOMOGENEOUS GROUND

Steven A. Arcone

INTRODUCTION

Good ground contacts are often necessary for the protection of equipment and personnel against excessive electrical transients or overloading. The idea is that when an excess of current is sent through the circuit a fault system activates to divert this current into the earth through suitably placed, low-resistance ground connections. The connections can be a simple arrangement of a metal rod placed in the ground or a more complicated arrangement of rods or buried wires. In every case, however, the resistivity of the earth materials encountered is an important design parameter.

In the Arctic, where permafrost or seasonally frozen ground is encountered, a knowledge of earth resistivity becomes all the more important for engineering purposes. Hoekstra et al. (1975) and Sellmann et al. (1974), using electromagnetic, noncontact methods of resistivity surveying, showed that resistivity in the Arctic is highly variable but that, with proper surveying techniques, suitable grounding sites can be located. Their work also showed that, in the Fairbanks area, values of earth resistivity are most commonly found between 100 and a few thousand ohm-meters over thawed and frozen sediments of varying ice content and that this range is often spanned in the active layer in the course of a year at any location. This is due to seasonal thaw and changes in the active layer depth, which must also be considered for grounding application.

Previously, Sunde (1949) and Tagg (1964) theoretically considered these grounding problems in non-homogeneous earth. However, Sunde did not pursue the effects of variations in layer thickness or the

effects of changing the penetration depth of a vertical rod. Tagg's analysis is unsuitable for rapid calculations because he presented an additional resistance due to the penetration of the rod into the lower earth layer that had to be added to the resistance of the portion of the rod in the upper layer. After these values were found, he presented a penetration factor which had to be multiplied by the sum of these resistances, making the calculation of resistance to ground laborious.

Since calculations of this type are so important to electrical engineering in the Arctic, a computer program was developed in this study for rapidly calculating the resistance to ground of either a vertical rod or a horizontal wire, both of specified length and cross section, for a one- or two-layered earth model. The program's accuracy was verified by comparing the results with those of Sunde (1949) for the simpler cases. No attempt was made to catalogue a series of earth-resistance curves for variations in all available parameters. The total computer program is presented in the appendix.

THEORY OF EARTH GROUNDING FOR SIMPLE ELECTRODES

Figure 1 shows an idealized case of a simple grounding configuration. A hemispherical electrode of radius r_0 inserted at the surface of an earth of resistivity ρ , measured in ohm-meters, delivers a current of I amperes to the ground. The homogeneity of the ground permits the current to flow symmetrically away from the electrode in hemispherical fashion such that the current density \vec{J} , measured in coulombs/(sec-m²), is




Figure 1. Hemispherical electrode discharging current into a homogeneous earth.

$$\vec{J} = \frac{I}{2\pi r^2} \hat{a}_r \quad (1)$$

where r is the radial distance from the electrode and \hat{a}_r is the radial unit vector.

The electric field E , measured in volts/meter, which determines the direction of current flow, is found from the relation

$$\vec{E} = \rho \vec{J} \quad (2)$$

and the electrostatic potential V , measured in volts at any distance r_1 , is defined as

$$V = \int_{r_0}^{r_1} \vec{E} \cdot d\vec{r} \quad (3)$$

where dr is the incremental radial distance of integration and the dot product is understood to be taken. Carrying out the integration, we find that

$$V = \frac{I\rho}{2\pi} \left(\frac{1}{r_0} - \frac{1}{r_1} \right). \quad (4)$$

For unit current delivered to the electrode, the resistance R to uniform ground out to a distance r_1 is then

$$R = \frac{V}{I} = \frac{\rho}{2\pi} \left(\frac{1}{r_0} - \frac{1}{r_1} \right) \text{ ohms.} \quad (5)$$

If the return electrode at r_1 is considered at infinity, then

$$R = \frac{\rho}{2\pi r_0} \quad (6)$$

is the resistance to ground of the hemispherical electrode.




Figure 2. Cylindrical coordinate system on a resistively layered earth used for solving Laplace's equation.

In cases where the earth may be idealized as two or more uniform layers of different resistivities, and the electrode geometry becomes more complicated, theoretical solutions can only be presented in integral or series form for which a computer must be used to obtain numerical answers.

The common procedure used was to solve Laplace's equation for the electrostatic potential

$$\nabla^2 V = 0 \quad (7)$$

in cylindrical coordinates, as illustrated in Figure 2. In the figure, the origin is taken at the surface of the earth, z is the depth below the surface, r is now the cylindrical radial distance, and ϕ is the angular coordinate. Laplace's equation must then be solved subject to certain constraints upon the current flow and the potential itself. The constraints are that, at the interface of any two layers, including that between air and earth, there must be a continuity of potential V and of normal current density J_z . Since, within each medium, the current density is defined as

$$\vec{J} = \vec{E}/\rho \quad (8)$$

where

$$\vec{E} = -\nabla V \quad (9)$$

all constraints can be mathematically expressed in terms of V . Air is considered infinitely resistive, so that $J_z = 0$ at $z = 0$.

MATHEMATICAL PROCEDURE

The approach used is based on the division of either electrode, the horizontal wire or vertical

rod,* into discrete segments. Considering one segment at a time, its electrostatic potential at every other segment position is computed. The potential at each segment is then the sum of the potentials derived from all the other segments. With unit current delivered to the entire rod, the resistance to ground is then the sum of all the segment potentials. This process demands n^2 calculations for n segments. To reduce this number, the total potential is well approximated by summing the potentials, calculated at the electrode centers, of all other segments.

Since the potential developed around each electrode segment is independent of the angular coordinate ϕ when the origin is placed in the center of that segment, eq 7 becomes

$$\frac{1}{r} \frac{\partial}{\partial r} \left(r \frac{\partial V}{\partial r} \right) + \frac{\partial^2 V}{\partial z^2} = 0 \quad (10)$$

when expressed in differential form. The solutions are stated without proof, as the general procedures for determining them are given in many texts (e.g., Stratton 1941, Ward 1967). All the solutions evaluated at the electrode center are of the integral form

$$V = \int_0^{\infty} f(\lambda, z, n, z_1, \rho_{1,2}, k_1) J_0(\lambda r) d\lambda \quad (11)$$

where λ = variable of integration

z = depth of the center of the rod

n = depth to the center of the segment in question

z_1 = depth to the two-layer interface

$\rho_{1,2}$ = resistivity of either layer

$k_1 = (\rho_1 - \rho_2) / (\rho_1 + \rho_2)$

J_0 = zero order Bessel function of the first kind

r = electrode radius

$d\lambda$ = incremental change in λ .

For the following cases, the term *segment* refers to that particular segment of the electrode whose potential we are calculating at the electrode center. Four different cases arise, depending on whether the center of the electrode or the center of the segment is in layer 1 or layer 2.

* The distinction between rod and wire is only on the basis of radii.

Case 1: Segment in layer 1, electrode center in layer 1:

$$V = \int_0^{\infty} \left(A e^{-\lambda z} + B e^{\lambda z} + \frac{\rho_1 I}{4\pi} \exp(-\lambda|z-n|) \right) J_0(\lambda r) d\lambda \quad (12)$$

where

$$A = -\frac{\rho_1 I}{4\pi} \left\{ \frac{k_1 \exp[-\lambda(2z_1-n)] - e^{-\lambda n}}{k_1 \exp(-2\lambda z_1) + 1} \right\} \quad (13)$$

$$B = -\frac{\rho_1 I}{4\pi} \left[\frac{k_1 \exp(-2\lambda z_1)(e^{-\lambda n} + e^{\lambda n})}{k_1 \exp(-2\lambda z_1) + 1} \right]. \quad (14)$$

Case 2: Segment in layer 1, electrode center in layer 2:

$$V = \int_0^{\infty} C e^{-\lambda z} J_0(\lambda r) d\lambda \quad (15)$$

where

$$C = \frac{\rho_1 I}{4\pi} (1 - k_1) \left[\frac{e^{-\lambda n} + e^{\lambda n}}{k_1 \exp(-2\lambda z_1) + 1} \right]. \quad (16)$$

Case 3: Segment in layer 2, electrode center in layer 1:

$$V = \int_0^{\infty} D (e^{-\lambda z} + e^{\lambda z}) J_0(\lambda r) d\lambda \quad (17)$$

where

$$D = \frac{\rho_1 I}{4\pi} (1 - k_1) \left[\frac{e^{-\lambda n}}{k_1 \exp(-2\lambda z_1) + 1} \right]. \quad (18)$$

Case 4: Segment in layer 2, electrode center in layer 2:

$$V = \int_0^{\infty} \left[\frac{\rho_2 I}{4\pi} \exp(-\lambda|z-n|) + E e^{-\lambda z} \right] J_0(\lambda r) d\lambda \quad (19)$$

where

$$E = \frac{\rho_2 I}{4\pi} e^{-\lambda n} \left\{ (1 + k_1) \left[\frac{1 + \exp(2\lambda z_1)}{k_1 \exp(-2\lambda z_1) + 1} \right] - \exp(2\lambda z_1) \right\}. \quad (20)$$




Figure 3. Resistance to ground as a function of electrode length for a vertical rod placed in various one- and two-layer earth models.




Figure 4. Resistance to ground as a function of electrode length for a horizontal wire lying on various one- and two-layer earth models.




Figure 5. Resistance to ground of a vertically driven rod as a function of first-layer thickness z .

These equations govern both horizontal and vertical electrode segments. For a horizontal electrode, either at the surface or buried, $z = n$. When a vertical electrode penetrates both layers, the following assumptions are used:

1. The current per unit length Y_1 in the portion of the electrode in the first layer, and the current per unit length Y_2 in the portion of the electrode in the second layer, are related to the total current I by the equation (Tagg 1964)

$$Y_1 L_1 + Y_2 L_2 = I \quad (21)$$

where L_1 and L_2 are the electrode lengths in the first and second layers, respectively.

2. Y_1 and Y_2 are related by the equation (Tagg 1964)

$$Y_1 = \frac{\rho_2}{\rho_1} Y_2 \quad (22)$$

The computer program developed, called RESIST, is written in BASIC computer language, and is listed in the appendix. The program performs the segmentation of the electrodes, integrates eq 12, 15, 17 and 19, and sums the resulting potentials to find the total resistance to ground for unit applied current. Either homogeneous ground or two-layer ground models can be considered. A definition of all the parameters used is listed at the start of the program.

RESULTS

To reduce the number of cases, only the following values were used: resistivity values of 100 and 1000 ohm-m, vertical rod radius of 0.05 m, and horizontal wire radius of 0.00125 m. These electrode dimensions were chosen to enable a comparison with Sunde's (1949) results for homogeneous ground to be made. The program considers the electrodes to be only at or below the surface.

The resistances to ground of the two types of electrodes, as a function of increasing electrode length for various one- and two-layer cases, are plotted in Figures 3 and 4. The dashed curves are Sunde's results and should be compared with the 100-ohm-m solid curves that are nearest them. These favorable comparisons ensure the validity of the program for wires less than 100 m long. The two-layer curves show the convergence of the resistance to ground of the one-layer and two-layer models for long rod lengths. The curves reveal that the horizontal wire gives superior grounding per meter of length when the upper layer is more conductive (100 ohm-m/1000 ohm-m case) than the lower layer, whereas the driven rod is superior when the lower layer is more conductive (1000 ohm-m/100 ohm-m case) than the upper layer. This is logical, since the better performance is exhibited by the electrode maintaining more contact with the 100-ohm-m earth.

The variations in resistance to ground as a function of first-layer thickness for the two different electrodes are plotted in Figures 5 and 6. This is an important case to consider for frozen ground applications where the active layer varies in thickness during the year. In areas of seasonal frost, where only the upper layer experiences freezing, the 30-m-long vertical rod electrode is superior to the 100-m-long wire for all layer thicknesses, as can be seen by




Figure 6. Resistance to ground of a horizontal wire resting on the surface as a function of first-layer thickness z_1 .

comparing the broken curves of the two figures. In permafrost regions, the 100-m-long wire is superior when the active layer is thawed, as can be seen by comparing the solid curves of the two figures. These results vary with electrode length, but they indicate the value of the program for considering these important situations.

PROBLEM AREAS

The following situations should be avoided when using the computer program listed in the appendix:

1. Vertical electrode (rod) radii less than 0.01 m.

The program approximates each segment of the electrode by its midpoint. This may become inaccurate with the present algorithm for electrode

division, which works to a tolerance of about 5% until the radius equals 0.01 m; radii smaller than this may therefore require more segments. Thus, it is recommended that vertical electrode radii be greater than 1 cm to save on computer editing and run time.

2. Center of the vertical electrode (rod) near the two-layer interface. This method of segmenting the electrode requires that each segment lie entirely within one layer. Therefore, if a desired electrode is found to have a center coincident with the modeled interface, it is recommended that it be lengthened a distance equal to six radii or lowered a distance of three radii to avoid this situation.

3. Horizontal electrodes (wires) buried beneath the surface. To check on the accuracy of the results for these cases, it is recommended that the electrostatic potential be recalculated for all segments where $r > 1$ m by decreasing $d\lambda$ by a factor of 10 or greater.

CONCLUSIONS

The program RESIST is an effective means of calculating the resistance to ground of simple electrodes. The number of variables allows a great deal of flexibility in dealing with two-layer earth models. Use of this program is facilitated by interactive aids whereby the user inputs data in response to formatted questions. When specific applications for arctic regions are concerned, seasonal resistivity and depth changes must be obtained for the site selected before a specific electrode type is decided upon. Then, curves such as those in Figures 5 and 6 can be used to determine the seasonal change of the ground resistance for specific electrode types so that a year-long grounding system can be designed.

LITERATURE CITED

- Hoekstra, P., P.V. Sellmann and A.J. Delaney (1975) Ground and airborne resistivity surveys of permafrost near Fairbanks, Alaska. *Geophysics*, vol. 40, no. 4, p. 641-656.
- Sellmann, P.V., P. Hoekstra, and A.J. Delaney (1974) Airborne resistivity survey: An aid in bedrock geology reconnaissance. U.S. Army Cold Regions Research and Engineering Laboratory (USA CRREL) Special Report 202. AD 777792.

- Stratton, J.A. (1941) *Electromagnetic theory*. New York: McGraw-Hill
- Sunde, E.D. (1949) *Earth conduction effects in transmission systems*. New York: D. Van Nostrand.
- Tagg, G.F. (1964) *Earth resistances*. New York: Pitman Publishing Corporation.
- Ward, S.H. (1967) *Mining geophysics*, vol. II. Tulsa, Oklahoma: The Society of Exploration Geophysicists.

RESIST

```

51 '          ***** DEFINITION OF PARAMETERS *****
52 '
53 '      A$ :  A CHARACTER VALUED VARIABLE WHICH CAN BE SET EQUAL
54 '            TO EITHER 'WIRE' OR 'ROD' - I.E., THE SOURCE OF
55 '            THE CURRENT CAN BE EITHER A VERTICALLY DRIVEN
56 '            ROD OR A WIRE LAID HORIZONTALLY.
57 '      L1 :  LENGTH OF THE ROD OR WIRE (METERS)
58 '      R1 :  RADIUS OF THE ROD OR WIRE (METERS)
59 '      I1 :  CURRENT IN THE ROD OR WIRE (AMPERES)
60 '      A :  NUMBER OF LAYERS IN THE SYSTEM (TWO IS THE MAXIMUM)
61 '      P1 :  RESISTIVITY OF THE UPPERMOST LAYER (OHM-METERS)
62 '      P2 :  RESISTIVITY OF THE SECOND LAYER (OHM-METERS)
63 '      Z1 :  DEPTH TO THE INTERFACE BETWEEN THE TWO LAYERS (METERS)
64 '      Z9 :  THE VERTICAL DISTANCE FROM THE SURFACE TO THE TOP OF
65 '            THE ROD (METERS). IF GREATER THAN ZERO, THEN THE
66 '            ROD IS BURIED.
67 '      Z :  DEPTH TO THE POINT AT WHICH THE POTENTIAL IS BEING
68 '            CALCULATED (METERS).
69 '      R :  RADIAL DISTANCE TO THE POINT AT WHICH THE POTENTIAL
70 '            IS BEING CALCULATED.
71 '      N :  DEPTH AT WHICH THE POINT SOURCE OF THE CURRENT IS
72 '            LOCATED (METERS). A VERTICALLY DRIVEN ROD IS
73 '            APPROXIMATED BY A FINITE NUMBER OR THESE POINT
74 '            SOURCES SUCH THAT N VARIES WHILE R REMAINS CON-
75 '            STANT. A WIRE IS DIVIDED SUCH THAT N IS CON-
76 '            STANT WHILE R VARIES.
77 '      L :  DUMMY VARIABLE OF INTEGRATION
78 '      K1 :  REFLECTION COEFFICIENT FOR A TWO-LAYERED SYSTEM,
79 '            EQUAL TO (P1-P2)/(P1+P2)
80 '      S :  RESISTANCE OF THE CURRENT SOURCE TO THE GROUND,
81 '            EQUAL TO THE SUM OF THE RESISTANCES OF EACH OF
82 '            ITS COMPOSITE POINT SOURCES.
83 '
95 REM OPTIONAL SHORTCUT
96 PRINT "DO YOU KNOW ENOUGH ABOUT THE PROGRAM'S OPERATION TO "
97 PRINT " TAKE THE SHORTCUT? IF SO, INPUT 1; IF YOU WANT THE "
98 PRINT " EXPLANATORY VERSION INPUT 2."
99 INPUT C9
100 IF C9=1 GO TO 110
101 IF C9=2 GO TO 400
102 PRINT "INCORRECT INPUT - TRY AGAIN"
104 GO TO 99
110 PRINT "INPUT A$,L1,R1,I1,A"
120 INPUT A$,L1,R1,I1,A
140 IF A=2 THEN 220
160 PRINT "INPUT P1"
180 INPUT P1
200 GO TO 260
220 PRINT "INPUT Z1,P1,P2"
240 INPUT Z1,P1,P2

```

RESIST (continued)

```

260 IF A$='WIRE' THEN 340
280 PRINT "INPUT Z9"
300 INPUT Z9
320 GO TO 1480
340 PRINT "INPUT N"
360 INPUT N
380 GO TO 1480
400 PRINT "ALL LENGTHS AND DISTANCES ARE TO BE INPUT IN METERS."
420 PRINT " AND RESISTANCE IS TAKEN TO BE MEASURED IN OHM-METERS."
440 PRINT "ALSO, THE ROD, OR WIRE, USED SHOULD NOT BE MUCH LESS"
480 PRINT " THAN ONE METER IN LENGTH."
500 PRINT "AVOID PLACING THE INTERFACE VERY CLOSE TO (OR AT) THE"
520 PRINT " MIDPOINT OF THE ROD, AS THIS MIGHT RESULT IN INACCU-"
540 PRINT " RATE OUTPUT."
560 PRINT
580 PRINT
600 PRINT "IS THIS A ONE- OR TWO-LAYER SYSTEM?"
620 INPUT A
640 PRINT
660 IF A=2 THEN 820
680 IF A=1 THEN 740
700 PRINT "INPUT '1' OR '2'."
720 GO TO 620
740 PRINT "WHAT IS THE RESISTANCE OF THE EARTH?"
760 INPUT P1
780 LET P2=P1
800 GO TO 920
820 PRINT "WHAT ARE THE RESISTIVITIES OF THE TWO LAYERS? (INPUT"
840 PRINT " THAT FOR THE UPPER LAYER FIRST.)"
860 INPUT P1,P2
880 PRINT "WHAT IS THE DEPTH TO THE INTERFACE BETWEEN THE TWO?"
900 INPUT Z1
920 PRINT "IS THE SOURCE OF THE CURRENT A STRETCHED OUT WIRE"
940 PRINT " (IF THIS IS THE CASE, TYPE 'WIRE') OR A DRIVEN ROD (IF"
960 PRINT " SO, TYPE 'ROD')?"
980 INPUT A$
1000 PRINT "WHAT IS THE TOTAL CURRENT IN THE SOURCE?"
1020 INPUT I1
1040 IF A$='ROD' THEN 1300
1060 IF A$='WIRE' THEN 1120
1080 PRINT "FORMAT ERROR. TYPE EITHER 'ROD' OR 'WIRE'."
1100 GO TO 980
1120 PRINT "WHAT IS THE LENGTH OF THE WIRE?"
1140 INPUT L1
1160 PRINT "AT WHAT DEPTH IS THE WIRE BURIED? IF IT LIES ON THE"
1180 PRINT " SURFACE, INPUT '0'."
1200 INPUT Z
1220 LET N=Z
1240 PRINT "WHAT IS THE RADIUS OF THE WIRE?"
1260 INPUT R1

```

RESIST (continued)

```

1280 GO TO 1480
1300 PRINT "WHAT IS THE RADIUS OF THE ROD?"
1320 INPUT R1
1340 PRINT "IF THE TOP OF THE ROD IS BENEATH THE SURFACE OF THE"
1360 PRINT " GROUND, THEN INPUT THE DISTANCE FROM THE SURFACE TO"
1380 PRINT " THE TOP OF THE ROD."
1400 INPUT Z9
1420 PRINT "WHAT IS THE LENGTH OF THE ROD?"
1440 INPUT L1
1460 PRINT
1480 '
1500 REM THIS SECTION CAN HANDLE THE SUBDIVISION OF EITHER A VERTICAL
1520 REM ROD OR A WIRE, STARTING AT THE MIDPOINT AND WORKING DOWN-
1540 REM WARD (OR OUTWARD).
1560 '
1580 DIM N(1000),I(1000)
1600 LET Z=L1/2+Z9 'THE MIDPOINT OF THE SOURCE
1620 IF A$="WIRE" THEN 1960
1640 IF A=1 THEN 1960
1660 '
1680 REM LINES 1760-2180 DETERMINE WHETHER OR NOT THE ROD CROSSES
1700 REM THE INTERFACE, AND, IF SO, DEFINE HOW THE CURRENT IN EACH
1720 REM SEGMENT WILL BE CALCULATED.
1740 '
1760 IF Z9<Z1 THEN 1800
1780 GO TO 1960
1800 IF L1+Z9>Z1 THEN 1840
1820 GO TO 1960
1840 LET L2=Z1-Z9 'LENGTH OF ROD IN UPPER LAYER
1860 LET Y2=I1/(P2/P1*L2+(L1-L2)) 'CURRENT PER UNIT LENGTH IN
1880 'THE SECOND MEDIUM.
1900 LET Y1=P2/P1*Y2 'CURRENT PER UNIT LENGTH IN
1920 'THE TOP LAYER.
1940 GO TO 2000
1960 LET Y9=1 'FOR FUTURE CHECKING; IT INDICATES THAT THE
1980 'ROD IS ENTIRELY WITHIN ONE MEDIUM.
2000 FOR N9=1 TO 5
2020 LET D=3*R1 'THE INITIAL SUBDIVISION SHOULD BE
2040 'NO LARGER THAN FOUR (4) TIMES THE RADIUS.
2060 IF A$="ROD" THEN 2120
2080 LET N(N9)=Z+D/2+(N9-1)*D
2100 GO TO 2200
2120 LET N(N9)=Z 'THIS PROGRAM SUBDIVIDES A ROD DIFFERENTLY THAN
2140 'IT DOES A WIRE. IT PUTS ONE SOURCE RIGHT AT THE
2160 'MIDPOINT, BUT ONE RADIUS AWAY (HORIZONTALLY) FROM
2180 'THE REFERENCE POINT.
2200 IF Y9=1 THEN 2800
2220 '
2240 REM LINES 2320-2780 AND LINES 3360-3640 HANDLE THE SUBDIVISION
2260 REM OF THE ROD AND THE CALCULATION OF CURRENT IN THE SEGMENTS

```

RESIST (continued)

```

2280 REM BELOW THE MIDPOINT WHEN IT EXTENDS ACROSS THE INTERFACE.
2300 '
2320 IF Z=Z1 THEN 2360
2340 GO TO 2440
2360 LET D=.001
2380 LET I(N9)=(Y1+Y2)/2*D
2400 LET D1=D
2420 GO TO 2740
2440 IF N(N9)>Z1 THEN 2520
2460 LET I(N9)=Y1*D
2480 IF N(N9)+D/2>Z1 THEN 2580
2500 GO TO 2820
2520 LET I(N9)=Y2*D
2540 IF N(N9)-D/2<Z1 THEN 2580
2560 GO TO 2820
2580 REM THIS IS THE SPECIAL CASE WHERE THE SEGMENT AROUND THE MIDPOINT
2600 REM OVERLAPS THE INTERFACE. A CORRECTION IS THEN NEEDED.
2620 LET D=2*ABS(Z1-Z)
2640 LET D1=D
2660 IF Z<Z1 THEN 2720
2680 LET I(N9)=Y2*D
2700 GO TO 2740
2720 LET I(N9)=Y1*D
2740 LET Q1=1
2760 REM (END OF SPECIAL CASE.)
2780 GO TO 2820
2800 LET I(N9)=(I1*D)/L1
2820 IF A$="ROD" THEN 3200 'FURTHER SUBDIVIDING UNNECESSARY
2840 LET S=D
2860 ' 'S' AND 'D' ARE USED TO KEEP TRACK OF THE LENGTHS OF THE SEG-
2880 ' MENTS. 'D' IS THE LENGTH OF THE CURRENT SEGMENT, AND 'S'
2900 ' IS THE LENGTH OF THE PREVIOUS ONE.
2960 IF N(N9)<(Z9+L1) THEN 3000
2980 GO TO 4040
3000 NEXT N9
3020 LET S=D
3040 LET D=2*D
3060 FOR N7=1 TO 5
3080 LET N9=N9+1 'KEEPS TRACK OF THE SEGMENTS
3100 LET N(N9)=(N(N9-1)+S/2)+D/2
3120 LET I(N9)=(I1*D)/L1
3140 LET S=D
3160 NEXT N7
3180 GO TO 3260
3200 LET S=D
3220 READ D
3240 GO TO 3280
3260 LET S=D
3280 READ D
3300 DATA .03,.05,.1,.2,.3,.4,.5,1,2,5,10,25,50

```

10

RESIST (continued)

```

3320 LET N9=N9+1
3340 LET N(N9)=(N(N9-1)+S/2)+D/2
3360 IF Y9=1 THEN 3660
3380 IF N(N9)>Z1 THEN 3440
3400 LET I(N9)=Y1*D
3420 GO TO 3720
3440 LET I(N9)=Y2*D
3460 IF N(N9-1)>Z1 THEN 3720
3480 IF Q1=3 THEN 3720
3500 REM CORRECTIONS FOR SEGMENTS OVERLAPPING THE INTERFACE.
3520 IF Q1=1 THEN 3700
3540 LET D=Z1-(N(N9-1)-S/2)
3560 LET N(N9-1)=Z1-D/2
3580 LET I(N9-1)=Y1*D
3600 LET N9=N9-1
3620 LET Q1=1
3640 GO TO 3720
3660 LET I(N9)=(I1*D)/L1
3680 GO TO 3720
3700 LET Q1=3
3720 IF N(N9)<((L1+Z9) THEN 3920
3740 IF Q1=3 THEN 4320
3760 IF N(N9)-N(N9-1)>>D THEN 4040
3780 GO TO 4180 'SHOULD GO THERE ONLY FOR VERY LONG WIRES, I.E.,
3800 'WHEN 'D' HAS REACHED '50'. REMEMBER THAT INIT-
3820 'IALLY IN THIS METHOD OF SUBDIVISION EACH SUCCESS-
3840 'IVE SEGMENT IS LONGER THAN THE PREVIOUS ONE.
3860 ' IN OTHER WORDS, N(N9)-N(N9-1)* D FOR RODS,
3880 'EXCEPT POSSIBLY IN LATER LINES, AFTER FINAL AD-
3900 'JUSTMENTS HAVE BEEN MADE.
3920 IF Q1=3 THEN 3960
3940 GO TO 3980
3960 LET Q1=0
3980 LET S=D
4000 IF D=50 THEN 3320
4020 GO TO 3260
4040 IF Y9=0 THEN 4100
4060 LET I(N9-1)=(I1*((L1+Z9)-(N(N9-1)-S/2)))/L1
4080 GO TO 4120
4100 LET I(N9-1)=(Y2*((L1+Z9)-(N(N9-1)-S/2)))
4120 LET N(N9-1)=((L1+Z9)+(N(N9-1)-S/2))/2
4140 LET N9=N9-1
4160 GO TO 4400
4180 IF Y9=0 THEN 4240 'NOT REALLY NEEDED
4200 LET I(N9-1)=(I1*((L1+Z9)-(N(N9-1)-D/2)))/L1
4220 GO TO 4260
4240 LET I(N9-1)=(Y2*((L1+Z9)-(N(N9-1)-D/2)))
4260 LET N(N9-1)=((L1+Z9)+(N(N9-1)-D/2))/2
4280 LET N9=N9-1
4300 GO TO 4400

```

RESIST (continued)

```

4320 REM THIS IS THE CASE WHERE THE INTERFACE IS NEAR THE BOTTOM
4340 REM OF THE ROD
4360 LET I(N9)=(Y2*((L1+Z9)-(N(N9)-D/2)))
4380 LET N(N9)=((L1+Z9)+(N(N9)-D/2))/2
4400 RESET
4420 IF A*="ROD" THEN 4500
4440 GO TO 6120 'FOR A WIRE, EVERYTHING IS SYMMETRIC
4460 'ABOUT THE MIDPOINT.
4480 '
4500 REM THIS SECTION SUBDIVIDES THE VERTICAL ROD, STARTING AT THE
4520 REM MIDPOINT AND WORKING UPWARD.
4540 '
4560 LET N9=N9+1
4580 IF D1>0 THEN 4640
4600 LET S=3*R1
4620 GO TO 4660
4640 LET S=D1
4660 READ D
4680 READ D
4700 LET N(N9)=(Z-S/2)-D/2
4720 IF Y9=1 THEN 5220
4740 '
4760 REM [LINES 4840-5200 AND LINES 5380-5660 HANDLE THE SUBDIVISION
4780 REM OF THE ROD AND THE CALCULATION OF CURRENT IN THE SEGMENTS
4800 REM ABOVE THE MIDPOINT WHEN IT EXTENDS ACROSS THE INTERFACE.
4820 '
4840 IF N(N9)>Z1 THEN 4900
4860 LET I(N9)=Y1*D
4880 GO TO 4940
4900 LET I(N9)=Y2*D
4920 GO TO 5260
4940 IF Z<Z1 THEN 5260
4960 IF Z=Z1 THEN 5180
4980 REM AGAIN, THIS IS THE SPECIAL CASE WHERE THE CENTER OF THE
5000 REM FIRST SEGMENT ABOVE THE MIDPOINT OVERLAPS THE INTERFACE.
5020 LET D=Z1-Z
5040 LET I(N9)=Y2*D
5060 LET N(N9)=Z-D/2
5080 LET S=D
5100 READ D
5120 LET Q1=2
5140 GO TO 5300
5160 REM (END OF SPECIAL CASE.)
5180 LET Q1=2
5200 GO TO 5260
5220 LET I(N9)=I1*D/L1
5240 GO TO 5280
5260 '
5280 LET S=D
5300 READ D

```

11

RESIST (continued)

```

5320 LET N9=N9+1
5340 LET N(N9)=(N(N9-1)-S/2)-D/2
5360 IF Y9=1 THEN 5680
5380 IF N(N9)>Z1 THEN 5440
5400 LET I(N9)=Y1*D
5420 GO TO 5480
5440 LET I(N9)=Y2*D
5460 GO TO 5740
5480 IF N(N9-1)<Z1 THEN 5740
5500 IF Q1=4 THEN 5740
5520 REM CORRECTION FOR SEGMENTS OVERLAPPING INTERFACE.
5540 IF Q1=2 THEN 5720
5560 LET D=(N(N9-1)+S/2)-Z1
5580 LET N(N9-1)=Z1+D/2
5600 LET I(N9-1)=Y2*D
5620 LET N9=N9-1
5640 LET Q1=2
5660 GO TO 5740
5680 LET I(N9)=(I1*D)/L1
5700 GO TO 5740
5720 LET Q1=4
5740 IF N(N9)>Z9 THEN 5800
5760 IF Q1=4 THEN 6040
5780 GO TO 5900 'THE CASE WHERE S=D DOES NOT EXIST FOR RODS
5800 IF Q1=4 THEN 5840
5820 GO TO 5860
5840 LET Q1=0
5860 LET S=D
5880 GO TO 5280
5900 IF Y9=0 THEN 5960
5920 LET I(N9-1)=(I1*((N(N9-1)+S/2)-Z9))/L1
5940 GO TO 5980
5960 LET I(N9-1)=(Y1*((N(N9-1)+S/2)-Z9))
5980 LET N(N9-1)=(Z9+(N(N9-1)+S/2))/2
6000 LET N9=N9-1
6020 GO TO 6120
6040 REM THIS IS THE CASE WHERE THE INTERFACE IS NEAR THE TOP
6060 REM OF THE ROD.
6080 LET I(N9)=(Y1*(N(N9)+D/2)-Z9)
6100 LET N(N9)=(Z9+(N(N9)+D/2))/2
6120 '
6140 FOR C=1 TO N9
6160 LET I=I+I(C)
6180 NEXT C
6200 IF I<I1+.00001 THEN 6240
6220 GO TO 6320
6240 IF I>I1-.00001 THEN 6420
6260 IF I>.5*I1-.00001 THEN 6300
6280 GO TO 6320
6300 IF I<.5*I1+.00001 THEN 6420

```

RESIST (continued)

```

6320 PRINT "THERE IS SOME ERROR IN THE SUBDIVISIONS. CHECK"
6340 PRINT " THE PROGRAM."
6360 PRINT
6380 PRINT "THE TOTAL CURRENT IN THE SOURCE IS";I;"AND IT SHOULD"
6400 PRINT " TOTAL";I1;"IF IT'S A ROD, OR";I1/2;"IF IT'S A WIRE."
6420 LET S=Q1=0
6440 REM END OF SUBDIVIDING
6460 PRINT
6480 RESET
6500 LET U=0
6520 DIM R(1000)
6540 LET D=.01
6560 LET X=200
6580 LET K1=(P1-P2)/(P1+P2)
6600 FOR C=1 TO N9
6620 LET I=I(C)
6640 IF A$="ROD" THEN 6760
6660 LET Z=N
6680 LET R9=L1/2
6700 LET R(C)=N(C)
6720 LET R=ABS(R9-R(C))
6740 GO TO 6800
6760 LET N=N(C)
6780 LET R=R1
6800 FOR L=0 TO X STEP D
6820 LET Y=L*R
6840 CALL "BESSEL";Y,J
6860 IF A=1 THEN 6900
6880 GO TO 7140
6900 IF N=0 THEN 7080
6920 '
6940 ' HERE THE EARTH IS ASSUMED TO BE HOMOGENEOUS
6960 CALL "1-LAYER";Q,L,P1,I,N,Z,J,K
6980 LET B$="1-LAYER"
7000 IF Q=2 THEN 7040
7020 GO TO 7580
7040 LET V=V+P1*I/(4*3.14159)*1/R
7060 GO TO 9080
7080 LET V=V+P1*I/(2*3.14159)*1/R 'WHEN THE WIRE IS ON THE SURFACE
7100 GO TO 9080
7120 '
7140 ' THE REMAINDER OF THE SUBPROGRAMS ASSUME THAT THE EARTH IS SEP-
7160 ' ARATED INTO TWO DISTINCT LAYERS OF SIGNIFICANTLY DIFFERENT
7180 ' RESISTIVITIES.
7200 '
7220 IF N>Z1 THEN 7260
7240 GO TO 7420
7260 IF Z>Z1 THEN 7300
7280 GO TO 7360
7300 CALL "S-2.MED2";Q,Q3,L,P1,P2,I,N,Z1,Z,K1,J,K

```

12

RESIST (continued)

```

7320 LET B$="S-2.MED2"
7340 GO TO 7580
7360 CALL "S-2.MED1";Q,L,P1,P2,I,N,Z1,Z,K1,J,K
7380 LET B$="S-2.MED1"
7400 GO TO 7580
7420 IF Z>Z1 THEN 7460
7440 GO TO 7520
7460 CALL "COEFF-C";Q,L,P1,I,N,Z1,Z,K1,J,K
7480 LET B$="COEFF-C"
7500 GO TO 7580
7520 CALL "COEFF";Q,Q3,L,P1,I,N,Z1,Z,R,K1,J,K
7540 LET B$="COEFF"
7560 IF Q3=4 THEN 8280 'NO NEED FOR INTEGRATION IN THIS CASE
7580 CALL "SIMPSON";Q,U,X,D,K,V
7600 LET U=U+1
7620 IF Q=1 THEN 7660
7640 NEXT L
7660 IF C>1 THEN 8280
7680 '
7700 ' IN RELATIVELY UNUSUAL CASES, THE USER MAY BE WARNED THAT
7720 ' GROSS INACCURACIES MIGHT EXIST. THE NECESSARY ALTER-
7740 ' ATIONS IN THE PROGRAM ARE LEFT TO THE USER'S DISCRETION.
7820 '
7840 IF A$="ROD" THEN 7880
7860 GO TO 8040
7880 IF R1<.01 THEN 7920
7900 GO TO 8280
7920 PRINT
7940 PRINT ' WARNING: FOR AN ACCURATE RESULT A THIN ROD SHOULD BE*
7960 PRINT ' DIVIDED INTO SMALL SEGMENTS ON EITHER SIDE OF THE*
7980 PRINT ' MIDPOINT. HOWEVER, THIS CAN LEAD TO SLOW CONVER-*
8000 PRINT ' GENCE AND A VERY LONG RUN TIME.*
8020 PRINT
8040 IF A$="WIRE" THEN 8080
8060 GO TO 8280
8080 IF N=0 THEN 8280
8100 PRINT
8120 PRINT ' WARNING: THE SIZE OF THE INTERVAL OF INTEGRATION BECOMES*
8140 PRINT ' INCREASINGLY IMPORTANT AS THE SOURCE MOVES AWAY FROM*
8160 PRINT ' THE MIDPOINT, AND VERY SMALL INTERVALS SHOULD BE USED*
8180 PRINT ' FOR THE OUTERMOST SEGMENTS.*
8200 PRINT
8260 '
8280 IF B$="S-2.MED2" THEN 8920
8300 IF B$="COEFF" THEN 8360
8320 GO TO 9080
8340 '
8360 REM REFER TO THE SUBPROGRAM 'COEFF'.
8380 IF Q3=4 THEN 8420 'WHEN Z=N=0
8400 GO TO 8460

```

RESIST (continued)

```

8420 LET V=K 'INTEGRATION NOT NEEDED, AS SERIES EXPANSION IS USED
8440 GO TO 9080
8460 IF Q3=9 THEN 9080 'WHEN COEFFICIENTS BECOME NEGLIGIBLE
8480 IF Q3=2 THEN 8860 'WHEN Z=N#Z1
8500 IF Q3=0 THEN 9080 'NO SHORTCUT
8520 IF Q3=5 THEN 8560
8540 GO TO 8800
8560 ' Q3 EQUALS 5 WHENEVER THE ABSOLUTE VALUE OF THE REFLECTION
8580 ' COEFFICIENT ((P1-P2)/(P1+P2)) IS CLOSE TO 1.0 (CUTOFF IS
8600 ' .99 , WHICH CORRESPONDS TO P1 BEING 5.0 AND P2 BEING 1000,
8620 ' FOR EXAMPLE). THIS MIGHT BE THE CASE IF ONE OF THE LAYERS
8640 ' IS SEA WATER, WHICH HAS A VERY LOW RESISTIVITY. WHEN BOTH
8660 ' THE SOURCE AND THE REFERENCE POINT ARE AT THE SURFACE, THE
8680 ' SERIES EXPANSION IS NORMALLY USED, BUT WHEN THE REFLECTION
8700 ' COEFFICIENT APPROACHES 1.00 OR -1.00 THE EXPANSION IS BOTH
8720 ' SLOW AND INACCURATE. THEREFORE THE ORIGINAL FUNCTION MUST
8740 ' BE EVALUATED AND INTEGRATED.
8760 LET V=V+P1*I/(4*3.14159)*(1+.28469)/R
8780 GO TO 9080
8800 'REACHES HERE WHEN Z=N=Z1
8820 LET U=V+P1*I/(4*3.14159)*(1-.28469*K1)*1/R
8840 GO TO 9080
8860 LET V=V+P1*I/(4*3.14159)*1/R
8880 GO TO 9080
8900 '
8920 REM REFER TO THE SUBPROGRAM 'S-2.MED2'.
8940 IF Q3=9 THEN 9080 'WHEN Z IS CLOSE TO N
8960 IF Q3=2 THEN 9060 'WHEN Z=N#Z1
8980 IF Q3=0 THEN 9080 'NO SHORTCUT
9000 'REACHES HERE IF Z=N=Z1
9020 LET V=V+P2*I/(4*3.14159)*.28469*(1/R+K1/R)
9040 GO TO 9080
9060 LET V=V+P2*I/(4*3.14159)*1/R
9080 LET S=S+V 'THE RESISTANCE (SUM OF ALL SEGMENTS)
9100 IF M7=3 THEN 9140
9120 PRINT 'LOCATION OF SOURCE RANGE OF INTEGRATION EFFECTIVE POTENTIAL'
9140 PRINT TAB(6);N(C);TAB(29);L;TAB(52);V
9160 LET M7=3
9180 LET U=Q=Q3=V=0
9200 NEXT C
9220 ' RELAYS THE RESULT TO THE USER
9240 IF A$="WIRE" THEN 9280
9260 GO TO 9300
9280 LET S=2*S
9300 PRINT
9320 PRINT "THE RESISTANCE OF THE ";A$;" TO THE GROUND IS";S;" , "
9340 PRINT " WHERE THE ";A$;" IS DIVIDED INTO";N9;" SEGMENTS WHOSE"
9360 PRINT " LENGTHS VARY WITH THEIR DISTANCE FROM THE MIDPOINT,"
9380 PRINT " WHICH IS LOCATED AT DEPTH";Z;" . "
9400 PRINT

```

RESIST (continued)

```

9420 LET S=V=Q=Q3=U=0
9440 PRINT *-----
9460 PRINT
9480 END
10000 SUB *EXPLAIN*:Y
10010 / THE FOLLOWING COMMENTS WILL BE HELPFUL FOR THE SUBPROGRAMS
10020 /   CONTAINED IN THIS FILE.
10030 /
10040 /   SUBPROGRAM *BESSEL*
10050 / THE VARIABLE 'Y' EQUALS L*XR, AND SERVES AS THE ARGUMENT OF THE
10060 / POLYNOMIAL THAT APPROXIMATES THE ZEROth ORDER BESSEL FUNCTION.
10070 / 'L' IS THE DUMMY VARIABLE OF INTEGRATION, AND 'R' IS THE HOR-
10080 / ZONTAL COMPONENT OF THE DISTANCE BETWEEN THE SOURCE OF THE
10090 / CURRENT AND THE POINT AT WHICH YOU WISH TO MEASURE THE PO-
10100 / TENTIAL.
10110 /
10120 /   SUBPROGRAM *SIMPSON*
10130 / THIS USES SIMPSON'S RULE TO INTEGRATE THE FUNCTION 'K'. THE
10140 / INPUTTED VALUES ARE: 1.) 'Q' - A CHECK TO PREVENT UNDERFLOW,
10150 / OVERFLOW, OR UNNECESSARY USE OF COMPUTER TIME; 2.) 'U' -
10160 / AN INTEGER WHICH INDICATES HOW MANY SUBINTERVALS HAVE ALREADY
10170 / BEEN CALCULATED; 3.) 'X' - THE UPPER LIMIT OF INTEGRATION,
10180 / WHICH IS THEORETICALLY EQUAL TO INFINITY WHILE A MUCH SMALLER
10190 / IS ALL THAT IS NEEDED FOR AN EXCELLENT APPROXIMATION; 4.)
10200 / 'D' - THE SIZE OF THE INTERVAL OF INTEGRATION (THERE SHOULD
10210 / BE ABOUT 100 OF THESE SUBINTERVALS PER UNIT 'L' - LAMBDA);
10220 / 5.) 'K' - ANY FUNCTION; 6.) 'V' - THE POTENTIAL.
10230 /
10240 /   SUBPROGRAM *COEFF*
10250 / THIS CALCULATES THE POTENTIAL FUNCTION FOR POINTS IN THE FIRST
10260 / LAYER WHEN THE SOURCE IS IN THE SAME LAYER.
10270 /
10280 /   SUBPROGRAM *COEFF-C*
10290 / CALCULATES THE POTENTIAL FUNCTION FOR POINTS IN THE LOWER OF
10300 / THE TWO LAYERS WHEN THE SOURCE IS IN THE UPPER LAYER.
10310 /
10320 /   SUBPROGRAM *S-2.MED1*
10330 / CALCULATES THE POTENTIAL FOR POINTS IN THE FIRST LAYER WHEN
10340 / THE SOURCE IS IN THE SECOND MEDIUM.
10350 /
10360 /   SUBPROGRAM *S-2.MED2*
10370 / BOTH THE SOURCE AND THE POINTS ARE IN THE LOWER LAYER.
10380 /
10390 / THE LATTER FOUR SUBPROGRAMS REQUIRE BASICALLY THE SAME
10400 / INPUT. THE FOLLOWING DESCRIPTIONS, THEN, REFER TO ALL FOUR.
10410 / 1.) 'Q' - A CHECK DEVICE TO PREVENT OVERFLOW, UNDERFLOW,
10420 / OR UNNECESSARY USE OF COMPUTER TIME; 2.) 'Q3' - (IN *COEFF*
10430 / AND *S-2.MED2* ONLY) - INDICATES TO THE USER AND TO THE COM-
10440 / PUTER WHICH SHORTCUTTING PATH IS FOLLOWED WITHIN A SUBPROGRAM
10450 / (THESE SHORTCUTS ARE DESIGNED TO SAVE COMPUTER TIME AND SPEED

```

14

RESIST (continued)

```

10460 / THE CONVERGENCE RATE OF THE POTENTIAL FUNCTIONS IN CERTAIN
10470 / PARTICULAR CASES, SUCH AS AT THE SURFACE, OR AT THE INTERFACE,
10480 / ETC.; 3.) 'L' - LAMBDA, THE DUMMY VARIABLE; 4.) 'P1' - THE
10490 / RESISTIVITY OF THE EARTH IN THE FIRST LAYER; 5.) 'P2' - (INPUT
10500 / ONLY FOR *S-2.MED1* AND *S-2.MED2*) - THE RESISTIVITY OF THE
10510 / EARTH IN THE THE SECOND LAYER; 6.) 'I' - THE TOTAL CURRENT
10520 / IN THE SOURCE; 7.) 'N' - THE DEPTH OF SOME PARTICULAR POINT
10530 / SOURCE (IT MAY BE THE CENTER OF A SEGMENT OF A LINE SOURCE.);
10540 / 8.) 'Z1' - THE DEPTH TO THE INTERFACE BETWEEN THE TWO LAYERS;
10550 / 9.) 'Z' - THE DEPTH (VERTICAL DISTANCE FROM THE SURFACE) AT
10560 / WHICH THE REFERENCE POINT IS LOCATED (FOR EXAMPLE, IN THE CASE
10570 / WHERE ONE IS CALCULATING THE RESISTANCE OF A DRIVEN ROD TO THE
10580 / GROUND, THIS REFERENCE POINT SHOULD BE AT THE SAME DEPTH AS THE
10590 / MIDPOINT OF THE ROD, BUT PLACED HORIZONTALLY ONE RADIUS AWAY.);
10600 / 10.) 'R' - THE RADIAL COMPONENT OF THE DISTANCE FROM THE SOURCE;
10610 / 11.) 'K1' - THE REFLECTION COEFFICIENT AT THE INTERFACE (IT'S
10620 / CALCULATED FROM THE RESISTIVITIES OF THE TWO MEDIA ON EITHER
10630 / SIDE.); 12.) 'J' - THE VALUE OF THE BESSEL FUNCTION, AS CALCULATED
10640 / AT EACH PARTICULAR LAMBDA FOR A GIVEN 'R'; 13.) 'K' - THE OUT-
10650 / PUTTED VALUE OF THE POTENTIAL FUNCTION, AS CALCULATED FROM THE
10660 / ELEVEN PARAMETERS JUST DESCRIBED.
10670 /
10680 /   SUBPROGRAM *1-LAYER*
10690 / THIS REPRESENTS THE SIMPLIFIED CASE WHERE THE EARTH IS HOMO-
10700 / GENEIOUS.
10710 / LSOURCE1, AND LSOURCE3.
10720 / IT'S PARAMETERS ARE THE SAME AS THOSE DESCRIBED FOR THE TWO-
10730 / LAYER CASE.
10740 /
10750 /
10760 SUBEND
10770 SUB *BESSEL*:Y,J
10780 IF Y>3 THEN 10900
10790 LET S1=-2.2499997*(Y/3)^2
10800 LET S2=1.2656208*(Y/3)^4
10810 LET S3=-.3163866*(Y/3)^6
10820 IF Y<.1 THEN 10870
10830 LET S4=.0444479*(Y/3)^8
10840 LET S5=-.0039444*(Y/3)^10
10850 LET S6=.00021*(Y/3)^12
10860 GO TO 10880
10870 LET S4=S5=S6=0
10880 LET J=1+S1+S2+S3+S4+S5+S6
10890 GO TO 10950
10900 LET F=.79788456-.00000077*(3/Y)-.0055274*(3/Y)^2-.00009512*(3/Y)^3
10910 LET F=F+.00137237*(3/Y)^4-.00072805*(3/Y)^5+.00014476*(3/Y)^6
10920 LET T=Y-.78539-.04166*(3/Y)-.00003954*(3/Y)^2+.00262573*(3/Y)^3
10930 LET T=T-.00054*(3/Y)^4-.00029333*(3/Y)^5+.00013558*(3/Y)^6
10940 LET J=Y^(-.5)*F*DCOS(T)
10950 SUBEND

```

RESIST (continued)

```

10960 '
10970 '
10980 SUB 'COEFF':Q,Q3,L,P1,I,N,Z1,Z,R,K1,J,K
10990 DEF FNG(Z)=A*EXP(-L*Z)+B*EXP(L*Z)
11000 DEF FNF(Z)=EXP(-L*ABS(Z-N))
11010 LET K9=P1*I/(4*3.14159)
11020 IF EXP(-2*L*Z1)<1E-35 THEN 11040
11030 GO TO 11130
11040 IF EXP(-L*N)<1E-5 THEN 11090
11050 LET A=-EXP(-L*N)
11060 LET B=0
11070 LET F1=1 'FLAG MEANING N<<Z1
11080 GO TO 11130
11090 IF EXP(-L*ABS(Z-N))<1E-2 THEN 11650
11100 LET Q3=9 'FLAG MEANING Z IS CLOSE TO N IN VALUE
11110 LET A=B=0
11120 IF Z=N THEN 11310
11130 IF Q3=9 THEN 11240
11140 IF F1>0 THEN 11170
11150 LET A=-(K1*EXP(-L*(2*Z1-N))-EXP(-L*N))/(K1*EXP(-2*L*Z1)+1)
11160 LET B=-K1*EXP(-2*L*Z1)*(EXP(-L*N)+EXP(L*N))/(K1*EXP(-2*L*Z1)+1)
11170 IF Z=N THEN 11310
11180 IF ABS(FNG(Z))<1E-5 THEN 11200
11190 GO TO 11220
11200 LET Q3=9
11210 GO TO 11000
11220 LET K=K9*(FNG(Z)+FNF(Z))*J
11230 GO TO 11260
11240 LET K=K9*FNF(Z)*J
11250 GO TO 11660
11260 IF ABS(Z-N)<.301 THEN 11290
11270 IF ABS(FNG(Z)+FNF(Z))<1E-3 THEN 11650
11280 GO TO 11660
11290 IF ABS(FNG(Z)+FNF(Z))<1E-2 THEN 11650
11300 GO TO 11660
11310 IF N=0 THEN 11430
11320 IF Z=Z1 THEN 11380
11330 LET K=K9*FNG(Z)*J
11340 IF ABS(FNG(Z))<1E-3 THEN 11360
11350 GO TO 11660
11360 LET Q3=2
11370 GO TO 11650
11380 LET K=K9*FNG(Z)*J
11390 IF L>4.999 THEN 11410
11400 GO TO 11660
11410 LET Q3=1
11420 GO TO 11650
11430 ' SPECIAL CASE WHERE BOTH THE CURRENT SOURCE AND THE REFERENCE
11440 ' POINT ARE AT THE SURFACE. THE EQUATION IS REDUCED TO A SERIES
11450 ' EXPANSION LIKE THAT ON PAGE 51 (EQ. 2.39) IN SUNDE (1949).

```

RESIST (continued)

```

11460 IF ABS(K1)>.99 THEN 11590
11470 ' WHEN THE RESISTIVITIES IF THE TWO LAYERS DIFFER SO THAT THE
11480 ' REFLECTION COEFFICIENT IS NEAR 1.0 IN ABSOLUTE VALUE, THE
11490 ' SERIES EXPANSION NO LONGER WORKS, AND THE ORIGINAL FUNCTION
11500 ' MUST BE EVALUATED AND INTEGRATED TO YIELD AN ACCURATE ANSWER.
11510 LET A=R/Z1
11520 LET X=10000
11530 FOR L=1 TO X
11540 LET S1=(-K1)^L/SQR(1+(2*L/A)^2)
11550 LET S=S+S1
11560 IF ABS(S1)<1E-6 THEN 11630
11570 NEXT L
11580 GO TO 11630
11590 LET K=K9*FNG(Z)*J
11600 IF L>4.999 THEN 11650
11610 LET Q3=5
11620 GO TO 11660
11630 LET Q3=4
11640 LET K=2*K9/R*(1+2*S)
11650 LET Q=1
11660 SUBEND
11670 '
11680 '
11690 SUB 'SIMPSON':Q,U,X,D,K,V
11700 REM INTEGRATION BY SIMPSON'S RULE
11710 IF U=0 THEN 11780
11720 IF U=X*(1/D) THEN 11780
11730 IF U/2=INT(U/2) THEN 11760
11740 LET I=4*K
11750 GO TO 11790
11760 LET I=2*K
11770 GO TO 11790
11780 LET I=K
11790 LET S=(D/3)*I
11800 LET V=V+S
11810 SUREND
11820 '
11830 '
11840 SUB 'COEFF-C':Q,L,P1,I,N,Z1,Z,K1,J,K
11850 DEF FNH(Z)=EXP(-L*(Z+N))+EXP(L*(N-Z))
11860 DEF FNJ(Z)=EXP(L*(N-Z))
11870 LET K9=P1*I/(4*3.14159)*(1-K1)
11880 IF EXP(-2*L*Z1)<1E-35 THEN 11900
11890 GO TO 11930
11900 IF EXP(-L*(Z+N))<1E-35 THEN 11970
11910 LET C=1
11920 GO TO 11940
11930 LET C=1/(K1*EXP(-2*L*Z1)+1)
11940 LET K=K9*C*FNH(Z)*J
11950 IF ABS(FNH(Z))<1E-3 THEN 12000

```

15

RESIST (continued)

```

11960 GO TO 12010
11970 LET K=K9*KC*FNJ(Z)*J
11980 IF ABS(FNJ(Z))<1E-3 THEN 12000
11990 GO TO 12010
12000 LET Q=1
12010 SUBEND
12020 '
12030 '
12040 SUB "S-2,MED1":Q,L,P1,P2,I,N,Z1,Z,K1,J,K
12050 DEF FNI(Z)=EXP(-L*(Z+N))+EXP(L*(Z-N))
12060 DEF FNR(Z)=EXP(L*(Z-N))
12070 LET K9=P1*I/(4*3.14159)*(1-K1)
12080 IF EXP(-L*(Z+N))<1E-35 THEN 12100
12090 GO TO 12120
12100 IF EXP(-2*L*Z1)<1E-10 THEN 12160
12110 GO TO 12190
12120 IF EXP(-2*L*Z1)<1E-10 THEN 12230
12130 GO TO 12250
12140 LET A=2*P2/(P1+P2)*EXP(-L*N)
12150 GO TO 12230
12160 LET A=1
12170 GO TO 12200
12180 GO TO 12270
12190 LET A=1/(K1*EXP(-2*L*Z1)+1)
12200 LET K=K9*A*FNR(Z)*J
12210 IF ABS(FNR(Z))<1E-3 THEN 12290
12220 GO TO 12300
12230 LET A=1
12240 GO TO 12260
12250 LET A=1/(K1*EXP(-2*L*Z1)+1)
12260 LET K=K9*A*FNI(Z)*J
12270 IF ABS(FNI(Z))<1E-3 THEN 12290
12280 GO TO 12300
12290 LET Q=1
12300 SUBEND
12310 '
12320 '
12330 SUB "S-2,MED2":Q,Q3,L,P1,P2,I,N,Z1,Z,K1,J,K
12340 LET K9=P2*I/(4*3.14159)
12350 IF Q3=9 THEN 12430
12360 LET D1=EXP(-L*N)
12370 LET D=2*P1/(P1+P2)*((1+EXP(2*L*Z1))/(K1*EXP(-2*L*Z1)+1))-EXP(2*L*Z1)
12380 LET D2=D*D1
12390 DEF FNJ(Z)=D2*EXP(-L*Z)
12400 IF Z=N THEN 12530
12410 DEF FNN(Z)=EXP(-L*ABS(Z-N))
12420 GO TO 12460
12430 LET K=K9*FNN(Z)*J
12440 IF EXP(-L*ABS(Z-N))<1 E-35 THEN 12670
12450 GO TO 12470

```

RESIST (continued)

```

12460 LET K=K9*(FNN(Z)+FNJ(Z))*J
12470 IF ABS(Z-N)<.301 THEN 12500
12480 IF ABS(FNJ(Z)+FNN(Z))<1E-3 THEN 12670
12490 GO TO 12510
12500 IF ABS(FNJ(Z)+FNN(Z))<1E-2 THEN 12670
12510 IF ABS(FNJ(Z))<1E-5 THEN 12650
12520 GO TO 12480
12530 IF Z=Z1 THEN 12580
12540 LET Q3=2
12550 LET K=K9*FNJ(Z)*J
12560 IF ABS(FNJ(Z))<1E-5 THEN 12670
12570 GO TO 12620
12580 LET Q3=1
12590 DEF FNK(Z)=(2*P1/(P1+P2))*((EXP(-2*L*Z)+1)/(K1*EXP(-2*L*Z)+1))
12600 LET K=K9*FNK(Z)*J
12610 IF L>4.999 THEN 12670
12620 IF EXP(-L*N)<1 E-35 THEN 12670
12630 IF EXP(-2*L*Z)<1 E-33 THEN 12670
12640 GO TO 12680
12650 LET Q3=9
12660 GO TO 12680
12670 LET Q=1
12680 SUBEND
12690 '
12700 '
12710 SUB "1-LAYER":Q,L,P1,I,N,Z,J,K
12720 LET K9=P1*I/(4*3.14159)
12730 IF Z=N THEN 12770
12740 IF EXP(-L*(Z+N))<1 E-35 THEN 12860
12750 DEF FNM(Z)=EXP(-L*ABS(Z-N))+EXP(-L*(Z+N))
12760 GO TO 12810
12770 DEF FNN(Z)=EXP(-2*L*Z)
12780 LET K=K9*FNN(Z)*J
12790 IF ABS(FNN(Z))<1E-3 THEN 12840
12800 GO TO 12870
12810 LET K=K9*FNM(Z)*J
12820 IF ABS(FNM(Z))<1E-3 THEN 12860
12830 GO TO 12870
12840 LET Q=2
12850 GO TO 12870
12860 LET Q=1
12870 SUBEND

```

U.S. GOVERNMENT PRINTING OFFICE: 1978-700-609/107